Zkratový proud alternátoru

Purpose

Průběh statorového a rotorového proudu při zkratu na synchronním alternátoru bez tlumiče.

System

Sites of Interaction

System Parameters

$U_n = 13.8$
$P_n = 115$
$S_n = 127.8$
$C_{osFi} = 0.9$
$I_n = 5350$
$I_{fn} = 1520$
$I_{f0} = 810$
$U_{fn} = 270$
$U_{f0} = 104$
$X_d = 116/100$
$X_q = 77/100$
$X_{dc} = 48/100$
$X_{dcc} = 29/100$
$X_{qcc} = 31/100$
$X_{sigma} = 22/100$
$X_2 = 30/100$
$R_s = 0.00494$
$R_f = 0.129$
$R_{vd} = 0.062$
$T_{d0c} = 7.1$
$T_{d0cc} = 2.033$
$T_{dc} = 2.3$
$T_{dcc} = 0.02$
$T_{q0cc} = 0.0373$
$T_{qcc} = 0.015$
$T_a = 0.28$
$O_{mega} = 314.15926$

$E_0 = 4$
$U_{q0} = 1$
$U_{d0} = 0.5$
$T_{heta0} = 0$
$I_{fu} = 1$

System excitation

Task

Assumptions

Solution

Model

Data

```
*: Zkratový proud alternátoru
*SYSTEM;

: Parametry alternatoru Dalesice dle skript PJS CVUT Cviceni
Un=13.8;
Pn=115;
Sn=127.8;
CosFi=0.9;
In=5350;
Ifn=1520;
If0=810;
Ufn=270;
Uf0=104;

: Reaktance
Xd=116/100;
Xq=77/100;
Xdc=48/100;
Xdcc=29/100;
Xqcc=31/100;
Xsigma=22/100;
X2=30/100;

: Cinne odpory
Rs=0.00494;
Rf=0.129;
Rvd=0.062;

: Casove konstanty
Td0c=7.1;
Td0cc=2.033;
```

```
Tdc=2.3;
Tdcc=0.02;
Tq0cc=0.0373;
Tqcc=0.015;
Ta=0.28;

Zn=Un*Un/Sn;
Omega=314.15926;
E0=4;
Uq0=1;
Ud0=0.5;

: Statorovy rozptyl je v obou osach priblizne stejny
Xdsigma=Xsigma;
Xqsigma=Xsigma;

: Zbyvajici reaktance a cinne odpory dle IdentifikaceParametruAlternatoru.doc
Xad=Xd-Xdsigma;
Xaq=Xq-Xqsigma;
Xfsigma=1/(1/(Xdc-Xdsigma)-1/Xad);
Xvdsigma=Xad*Xfsigma*(Xdsigma-Xdcc)/(Xad*(Xdcc-Xdsigma-Xfsigma)+Xfsigma*(Xdcc-Xdsigma));
Xvqsigma=1/(1/(Xqcc-Xqsigma)-1/Xaq);

Xf=Xfsigma+Xad;
Tf=(Xf*Zn/Omega)/Rf;
Xfc=Xf/Xd*Xdc;
Tfc=(Xfc*Zn/Omega)/Rf;

: Komponenty proudu Id
Idu=-E0/Xd;
Ida=-Uq0*(1/Xdc-1/Xd);
Idomegam=(Uq0*cos(Omega*Time)+Ud0*sin(Omega*Time))/Xdc;
Id=Idu + Ida*exp(-Time/Tfc) + Idomegam*exp(-Time/Ta);
: Komponenty proudu Iq
Iqomegam=(Uq0*sin(Omega*Time)-Ud0*cos(Omega*Time))/Xq;
Iq=Iqomegam*exp(-Time/Ta);

: Statorov0 proudy
Theta0=0;
Ia=Id*cos(Omega*Time+Theta0)      +Iq*sin(Omega*Time+Theta0);
Ib=Id*cos(Omega*Time+Theta0-2pi/3)+Iq*sin(Omega*Time+Theta0-2pi/3);
Ic=Id*cos(Omega*Time+Theta0-2pi/3)+Iq*sin(Omega*Time+Theta0+2pi/3);

: Rotorove proudy
Ifu=1;
Ifa=-Ida*Xd/Xad;
Ifomega=-Ifa;
If=Ifu + Ifa*exp(-Time/Tfc) + Ifomega*exp(-Time/Ta)*cos(Omega*Time);

*TR;
TR 0 0.4;
PRINT(4001) Idu, Ida, Idomegam,
            Iqomegam, Id, Iq,
            Ia, Ib, Ic,
            Ifu, Ifa, Ifomega, If;
RUN;
```

```
*END;
```

Statorové proudy v ose "d".



Statorové proudy v ose "q".



Statorové proudy výsledné v osách "d" a "q".



Statorové proudy fázové v osách "a", "b" a "c".



Rotorové proudy.



Origin

Karel Noháč KEE, FEL, ZČU v Plzni

Last Update

November 26, 2013